/*
 * antlr.gradle
 *
 * This source file is part of the FoundationDB open source project
 *
 * Copyright 2015-2024 Apple Inc. and the FoundationDB project authors
 *
 * Licensed under the Apache License, Version 2.0 (the "License");
 * you may not use this file except in compliance with the License.
 * You may obtain a copy of the License at
 *
 *     http://www.apache.org/licenses/LICENSE-2.0
 *
 * Unless required by applicable law or agreed to in writing, software
 * distributed under the License is distributed on an "AS IS" BASIS,
 * WITHOUT WARRANTIES OR CONDITIONS OF ANY KIND, either express or implied.
 * See the License for the specific language governing permissions and
 * limitations under the License.
 */

apply plugin: 'antlr'

generateGrammarSource {
    outputDirectory = layout.buildDirectory.dir('generated-src/antlr/main/com/apple/foundationdb/relational/generated').get().asFile
    maxHeapSize = "128m"
    arguments += ['-package','com.apple.foundationdb.relational.generated', '-listener','-visitor', '-long-messages']
}

sourceSets.configureEach {
    var generateGrammarSource = tasks.named(getTaskName("generate", "GrammarSource"))
    java.srcDir(generateGrammarSource.map { files() })
}

generateGrammarSource.doLast {
    final source = layout.buildDirectory.dir('generated-src/antlr/main/com/apple/foundationdb/relational/generated/')
    final tokensFile = "*.tokens"
    final dest = layout.projectDirectory.dir('src/main/antlr')
    copy {
        from source
        include tokensFile
        into dest
    }
}

task cleanUpTokenFiles(type: Delete) {
    final tokenFilesDir = layout.projectDirectory.dir('src/main/antlr')
    delete fileTree(tokenFilesDir) { include '*.tokens' }
}

clean {
    dependsOn "cleanUpTokenFiles"
}